

```

FINISH
/CLEAR
C*****
C***
C*** RECIPIENTE IN PRESSIONE IN PARETE SOTTILE
C***
C*** ELEMENTI GUSCIO ASSIALSIMMETRICO
C***
C*** PARAMETRI
/PREP7
D=1800      ! diametro recipiente
L=4000      ! lunghezza recipiente (fondi esclusi)
HF=400      ! semi-asse minore fondi
SP=3        ! spessore
PINT=1      ! pressione interna
NC=50       ! numero elementi fasciame cilindrico
NR=10       ! numero elementi zona raccordo
NF=40       ! numero elementi fondo
RE=4        ! rapporto dimensioni elementi
RSP=5       ! dimensioni zona infittita giunzione/spessore
C***
C*** NODI
C***
N,1,D/2
N,NC+1,D/2,L/2-SP*RSP
FILL,1,NC+1,,,,,1/RE
NGEN,NR+1,1,NC+1,NC+1,,0,SP*RSP/NR
LOCAL,11,1,0,L/2,0,,,,HF/(D/2)
NGEN,NR+1,1,NC+NR+1,,,0,((SP*RSP)/HF/3.14*180)/NR
N,NC+2*NR+NF+1,D/2,90
FILL,NC+2*NR+1,NC+2*NR+NF+1,,,,,RE
CSYS,0
C***
C*** MATERIALE
C***
MP,EX,1,210000
C***
C*** ELEMENTI
C***
ET,1,51
R,1,SP
E,1,2
EGEN,NC+2*NR+NF,1,-1
C***
C*** VINCOLI
C***
D,1,UY,0
D,1,ROTZ,0
D,1,UZ,0
D,NC+2*NR+NF+1,UX,0
D,NC+2*NR+NF+1,ROTZ,0
C***
C*** CARICHI
C***
SFE,ALL,1,PRES,,PINT
FINISH
C***
C*** SOLUZIONE
C***
/SOLU
SOLVE
FINISH
C***

```

```

C*** POST-PROCESSING
C***
/POST1
C*** opzioni di rappresentazione
/PBC,ALL,,1
/PBC,NFOR,,0
/PBC,NMOM,,0
/PBC,RFOR,,0
/PBC,PATH,,0
/PNUM,NODE,0
/PSF,PRES,NORM,2
C*** struttura con vincoli e carichi
/TITLE,MODELLO
EPLOT
NSEL,,NODE,,1
NSEL,A,NODE,,NC+1
NSEL,A,NODE,,NC+NR+1
NSEL,A,NODE,,NC+2*NR+1
NSEL,A,NODE,,NC+2*NR+NF+1
/PNUM,NODE,1
/NOERASE
NPLOT
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0

/ERASE
/PNUM,NODE,0
/PSF,PRES,NORM,0
NSEL,ALL
/TITLE,NODI
NPLOT
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0

C*** deformata
/TITLE,DEFORMATA
PLDISP,1
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0
C*** sforzo normale
/PBC,ALL,,0
ETABLE,SLT,LS,1      ! estrae il dato "tensione longitudinale" (TOP)
ETABLE,SLM,LS,5      ! MID
ETABLE,SLB,LS,9      ! BOTTOM
ETABLE,SCT,LS,3      ! estrae il dato "tensione circonferenziale" (TOP)
ETABLE,SCM,LS,7      ! MID
ETABLE,SCB,LS,11     ! BOTTOM
ETABLE,STT,LS,2      ! estrae il dato "tensione taglio spessore" (TOP)
ETABLE,STM,LS,6      ! MID
ETABLE,STB,LS,10     ! BOTTOM

SADD,SLF,SLT,SLM,1,-1 ! calcola la tensione flessionale longitudinale
SADD,SCF,SCT,SCM,1,-1 ! calcola la tensione flessionale circonferenziale

/TITLE,TENSIONE LONGITUDINALE (MEMBRANALE)
PLLS,SLM,SLM
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0

/TITLE,TENSIONE LONGITUDINALE (FLESSIONALE)
PLLS,SLF,SLF
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0

/TITLE,TENSIONE CIRCOFERENZIALE (MEMBRANALE)
PLLS,SCM,SCM
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0

```

```
/TITLE,TENSIONE CIRCONFERENZIALE (FLESSIONALE)  
PLLS,SCF,SCF  
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0
```

```
/TITLE,TENSIONE DI TAGLIO NELLO SPESSORE  
PLLS,STM,STM ! rappresenta il taglio  
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0
```